
第 2章

アセンブリ言語譜

; ===========================================================
; CP/M ADAPTER ver 1.0H
; ===========================================================

; -----------------------------------------------------------
; FPGAポート
; -----------------------------------------------------------
CNTP EQU 0 ; IN 制御ポート
RECVP EQU 1 ; IN 受信ポート
SENDP EQU 2 ; OUT 送信ポート
REQP EQU 3 ; OUT 要求ポート

; CNTP.0 1: 受信データあり
; CNTP.1 1: 送信中
; REQP.0 1: FPGAの送信を停止

; -----------------------------------------------------------
; 0番地 WBOOT 分岐
; -----------------------------------------------------------

ORG 0H

JP BIOSENTRY+3

; -----------------------------------------------------------
; 3番地 IOBYTE
; -----------------------------------------------------------

ORG 3H

DB 00,00

; -----------------------------------------------------------
; 5番地 BDOS 分岐
; -----------------------------------------------------------

ORG 5H

JP SYSCALL

; -----------------------------------------------------------
; BDOS ENTRY
; -----------------------------------------------------------

ORG 0F000H
BDOSENTRY:
SYSCALL:

LD A,C

CP 0
JR Z,SYSCALL0

CP 1

JR Z,SYSCALL1

CP 2
JR Z,SYSCALL2

CP 5
JR Z,SYSCALL5

CP 9
JP Z,SYSCALL9

CP 10
JR Z,SYSCALL10

CP 11
JR Z,SYSCALL11

CP 12
JR Z,SYSCALL12

LD (ERRNO),A

JP ERROR

SYSCALL0: JP WBOOT
SYSCALL1: JP CONIN
SYSCALL2:

LD C,E
JP CONOUT

SYSCALL5:
LD C,E
JP CONOUT

SYSCALL10: JP CONBUFREAD
SYSCALL11: JP CONST
SYSCALL12: JP CPMVER

; -----------------------------------------------------------
; BIOS ENTRY
; -----------------------------------------------------------

ORG 0F100H
HEAD:
BIOSENTRY:

JP BOOT
JP WBOOT
JP CONST
JP CONIN
JP CONOUT
JP LIST
JP PUNCH
JP READER
JP HOME
JP SELDSK
JP SETTRK
JP SETSEC
JP SETDMA
JP READ
JP WRITE
JP LISTST
JP SECTRAN

; -----------------------------------------------------------
; 初期設定
; -----------------------------------------------------------
SYSRESET:
BOOT:

WBOOT:
LD A,0
LD (EXEFLG),A
LD (ECHOST),A
LD (CMDNO),A
OUT (REQP),A

LD SP,0FFFFH
; -----------------------------------------------------------
; 開始文字列表示
; -----------------------------------------------------------

LD HL,MESSAGE
SYSRESET0:

LD A,(HL)
CP "$"
JR Z,CPP0

LD C,2
LD E,A
CALL SYSCALL

INC HL
JR SYSRESET0

; -----------------------------------------------------------
; コマンド入力待ち表示
; -----------------------------------------------------------
CPP0:
; LD E,0DH
; LD C,2
; CALL SYSCALL

; LD E,"%"
; LD C,2
; CALL SYSCALL
; -----------------------------------------------------------
; 入力待ち
; -----------------------------------------------------------

LD HL,BUF
LD B,0

CPP1:
LD C,1
CALL SYSCALL

;
; 入力文字をメモリに記憶
;

LD (HL),A
;
; 文字数の上限を検査
;

INC HL
INC B
LD C,A
LD A,B
CP 15
JR NC,CPP0
LD A,C

;
; 改行検査
;

CP 0DH
JR NZ,CPP1

; -----------------------------------------------------------
; コマンド文字列の GO と比較
; -----------------------------------------------------------

LD HL,CMD0
LD IX,BUF



6 第 2章 アセンブリ言語譜

CPP1A:
LD A,(HL)
LD D,(IX)
CP "$"
JR Z,CPP1A0 ; コマンド文字と一致
CP D
JR NZ,NEXTCMD0 ; コマンド文字と不一致
INC HL
INC IX
JR CPP1A

CPP1A0:
LD A,1
LD (CMDNO),A
JR CPP2

; -----------------------------------------------------------
; コマンド文字列の GOE と比較
; -----------------------------------------------------------
NEXTCMD0:

LD HL,CMD1
LD IX,BUF

CPP1B:
LD A,(HL)
LD D,(IX)
CP "$"
JR Z,CPP1B0 ; コマンド文字と一致
CP D
JR NZ,CPP2 ; コマンド文字と不一致
INC HL
INC IX
JR CPP1B

CPP1B0:
LD A,2
LD (CMDNO),A
JR CPP2

; -----------------------------------------------------------
; コマンドに指定される 16ビットの番地を取り出す
; -----------------------------------------------------------
CPP2:

PUSH IX
POP HL
LD A,(HL)
LD DE,0
LD (NO16BIT),DE

CPP2A:
CP "0"
JR C,CPP0 ; 文字が 16進数以外
CP "9"+1
JR C,CPP3 ; 文字が 0から 9
CP "A"
JR C,CPP0 ; 文字が 16進数以外
CP "F"+1
JR C,CPP4 ; 文字が Aから F
JR CPP0 ; 文字が 16進数以外

CPP3: ; 文字の 0から 9を数値に変換
SUB "0"
JR CPP5

CPP4: ; 文字の Aから Fを数値に変換
SUB "A"
ADD A,10

CPP5:
PUSH HL
LD HL,NO16BIT
RLD
LD HL,NO16BIT+1

RLD
POP HL

INC HL
LD A,(HL)
CP 0DH
JR Z,EXECMD ; 文字から数値を取り出し終了

JR CPP2A ; 文字から数値の取り出し継続

; -----------------------------------------------------------
; コマンド実行
; -----------------------------------------------------------
EXECMD:

LD A,(CMDNO)
CP 1
JR Z,CMDGO
CP 2
JR Z,CMDGOE

JP CPP0

; -----------------------------------------------------------
; GO コマンド
; -----------------------------------------------------------
CMDGO:

LD A,0
LD (ECHOST),A
LD A,1
LD (EXEFLG),A
OUT (REQP),A

LD HL,(NO16BIT)
JP (HL)

; -----------------------------------------------------------
; GOE コマンド
; -----------------------------------------------------------
CMDGOE:

LD A,1
LD (EXEFLG),A
LD (ECHOST),A
OUT (REQP),A

LD HL,(NO16BIT)
JP (HL)

; -----------------------------------------------------------
; SYSTEM CALL 9: 文字列のプリントアウト
; -----------------------------------------------------------
SYSCALL9:

LD H,D
LD L,E

SYSCALL9A:
LD A,(HL)
CP "$"
JR Z,SYSCALL9B

LD C,2
LD E,A
CALL SYSCALL

INC HL
JR SYSCALL9A

SYSCALL9B:
RET



8 第 2章 アセンブリ言語譜

; -----------------------------------------------------------
; BIOS CONIN
; -----------------------------------------------------------
CONIN:

CALL ECHOON
IN A,(CNTP)
AND 1
JR Z,CONIN
IN A,(RECVP)
CALL ECHOOFF
RET

; -----------------------------------------------------------
; BIOS CONOUT
; -----------------------------------------------------------
CONOUT:

IN A,(CNTP)
AND 2
JR NZ,CONOUT
LD A,C
OUT (SENDP),A
RET

; -----------------------------------------------------------
; BIOS CPMVER
; -----------------------------------------------------------
CPMVER:

LD H,0
LD L,22H
RET

; -----------------------------------------------------------
; BIOS CONST
; -----------------------------------------------------------
CONST:

CALL ECHOON
IN A,(CNTP)
AND 1
JR Z,CONST0
LD A,0FFH

CONST0:
CALL ECHOOFF
RET

; -----------------------------------------------------------
; BIOS CONBUFREAD
; -----------------------------------------------------------
CONBUFREAD:

CALL ECHOON
LD A,(DE)
LD (BUFSIZE),A
INC DE
LD H,D
LD L,E
INC DE
LD C,0

CONBUFREAD0:
IN A,(CNTP)
AND 1
JR Z,CONBUFREAD0

IN A,(RECVP)
LD (DE),A
INC DE

INC C

CP 0DH
JR Z,CONBUFREAD1

LD A,(BUFSIZE)
CP C
JR NC,CONBUFREAD0
LD A,C
CP 0FFH
JR NZ,CONBUFREAD0

CONBUFREAD1:
DEC DE
LD A,0
LD (DE),A
DEC C
LD (HL),C
CALL ECHOOFF
RET

; -----------------------------------------------------------
; BIOS 未定義番号
; -----------------------------------------------------------
LIST:

LD A,2
LD (NOBIOSNO),A
RET

PUNCH:
LD A,3
LD (NOBIOSNO),A
RET

READER:
LD A,4
LD (NOBIOSNO),A
RET

HOME:
LD A,5
LD (NOBIOSNO),A
RET

SELDSK:
LD A,6
LD (NOBIOSNO),A
RET

SETTRK:
LD A,7
LD (NOBIOSNO),A
RET

SETSEC:
LD A,8
LD (NOBIOSNO),A
RET

SETDMA:
LD A,9
LD (NOBIOSNO),A
RET

READ:
LD A,10
LD (NOBIOSNO),A

RET

WRITE:
LD A,11
LD (NOBIOSNO),A
RET

LISTST:
LD A,12
LD (NOBIOSNO),A
RET

SECTRAN:
LD A,13
LD (NOBIOSNO),A
RET

; -----------------------------------------------------------
; 未定義処理呼び出しの記憶
; -----------------------------------------------------------
ERROR:

CP 11H
JR Z,ERROR2
CP 0FH
JR Z,ERROR5
CP 1AH
JR Z,ERROR6
CP 14H
JR Z,ERROR7
CP 19H
JR Z,ERROR8

LD HL,ERRMESSAGE
ERROR0:

LD A,(HL)
CP "$"
JR Z,ERROR1

LD C,2
LD E,A
CALL SYSCALL

INC HL
JR ERROR0

ERROR1:
JP SYSRESET

ERROR2:
LD HL,ERRMESSAGE1
JR ERROR3

ERROR5:
LD HL,ERRMESSAGE2
JR ERROR3

ERROR6:
LD HL,ERRMESSAGE3
JR ERROR3

ERROR7:
LD HL,ERRMESSAGE4
JR ERROR3

ERROR8:
LD HL,ERRMESSAGE5

ERROR3:
LD A,(HL)
CP "$"
JR Z,ERROR4

LD C,2



11

LD E,A
CALL SYSCALL

INC HL
JR ERROR3

ERROR4:
LD A,0
RET

; -----------------------------------------------------------
; エコーオン
; -----------------------------------------------------------
ECHOON:

PUSH AF
LD A,(ECHOST)
CP 1
JR NZ,ECHOON0
LD A,0
OUT (REQP),A

ECHOON0:
POP AF
RET

; -----------------------------------------------------------
; エコーオフ
; -----------------------------------------------------------
ECHOOFF:

PUSH AF
LD A,(ECHOST)
CP 1
JR NZ,ECHOOFF0
LD A,1
OUT (REQP),A

ECHOOFF0:
POP AF
RET

; -----------------------------------------------------------
; 変数
; -----------------------------------------------------------
BUF: DS 16
NO16BIT: DS 2
ERRNO: DS 1
EXEFLG: DS 1
NOBIOSNO: DS 1
CMDNO: DS 1
BUFSIZE: DS 1
ECHOST: DS 1

; -----------------------------------------------------------
; 定数と文字列
; -----------------------------------------------------------
FREEZMESSAGE: DB 0DH,"CPU HALT",0DH,0AH,"$"
ERRMESSAGE: DB 0DH,"ERROR",0DH,0AH,"$"
ERRMESSAGE1: DB 0DH,"system call no.17 unsupport",0DH,0AH,"$"
ERRMESSAGE2: DB 0DH,"system call no.15 unsupport",0DH,0AH,"$"
ERRMESSAGE3: DB 0DH,"system call no.26 unsupport",0DH,0AH,"$"
ERRMESSAGE4: DB 0DH,"system call no.20 unsupport",0DH,0AH,"$"
ERRMESSAGE5: DB 0DH,"system call no.25 unsupport",0DH,0AH,"$"
MESSAGE: DB 0DH,0AH,"CP/M ADAPTER ver 1.0H",0DH,0AH,"$"
CMD0: DB "GO ","$"
CMD1: DB "GOE ","$"

; -----------------------------------------------------------
; BDOS先頭までを 0にする
; -----------------------------------------------------------

ORG 0F800H

LD HL,00H
FILL:

LD A,0
LD (HL),A
LD A,0FFH
CP L
JR NZ,FILL0
LD A,0EFH
CP H
JR NZ,FILL0

LD A,0C3H
LD (0),A
LD HL,BIOSENTRY+3
LD A,L
LD (1),A
LD A,H
LD (2),A

LD A,0C3H
LD (5),A
LD HL,BDOSENTRY
LD A,L
LD (6),A
LD A,H
LD (7),A

LD A,0
LD (3),A
LD (4),A

JP SYSRESET
FILL0:

INC HL
JR FILL

; -----------------------------------------------------------
; CPUを停止
; -----------------------------------------------------------

ORG 0F900H

LD HL,FREEZMESSAGE
FREEZ:

LD A,(HL)
CP "$"
JR Z,FREEZ0

LD C,2
LD E,A
CALL SYSCALL

INC HL
JR FREEZ

FREEZ0:
LD A,0FFH
LD (EXEFLG),A
LD A,0
OUT (REQP),A
HALT

FREEZ1:
JR FREEZ1

end

第 3章

割付番地

F000 BDOSENTRY F100 BIOSENTRY F133 BOOT
F330 BUF F346 BUFSIZE F412 CMD0
F416 CMD1 F1EF CMDGO F1FF CMDGOE
F345 CMDNO F24B CONBUFREAD F258 CONBUFREAD0
F272 CONBUFREAD1 F21E CONIN F22D CONOUT
F23C CONST F247 CONST0 F237 CPMVER
F154 CPP0 F159 CPP1 F173 CPP1A
F183 CPP1A0 F191 CPP1B F1A1 CPP1B0
F1A8 CPP2 F1B3 CPP2A F1C5 CPP3
F1C9 CPP4 F1CD CPP5 F322 ECHOOFF
F32E ECHOOFF0 F314 ECHOON F320 ECHOON0
F347 ECHOST F354 ERRMESSAGE F35D ERRMESSAGE1
F37C ERRMESSAGE2 F39B ERRMESSAGE3 F3BA ERRMESSAGE4
F3D9 ERRMESSAGE5 F342 ERRNO F2C4 ERROR
F2DB ERROR0 F2E9 ERROR1 F2EC ERROR2
F303 ERROR3 F311 ERROR4 F2F1 ERROR5
F2F6 ERROR6 F2FB ERROR7 F300 ERROR8
F1E1 EXECMD F343 EXEFLG F803 FILL
F83B FILL0 F903 FREEZ F911 FREEZ0
F91B FREEZ1 F348 FREEZMESSAGE F100 HEAD
F28E HOME F27C LIST F2B8 LISTST
F3F8 MESSAGE F18A NEXTCMD0 F340 NO16BIT
F344 NOBIOSNO F282 PUNCH F2AC READ
F288 READER F2BE SECTRAN F294 SELDSK
F2A6 SETDMA F2A0 SETSEC F29A SETTRK
F000 SYSCALL F028 SYSCALL0 F02B SYSCALL1
F036 SYSCALL10 F039 SYSCALL11 F03C SYSCALL12
F02E SYSCALL2 F032 SYSCALL5 F20D SYSCALL9
F20F SYSCALL9A F21D SYSCALL9B F133 SYSRESET
F146 SYSRESET0 F133 WBOOT F2B2 WRITE

